# Maintainer: Antonio Rojas <arojas@archlinux.org>

pkgname=python-database-knotinfo
_pyname=database_knotinfo
pkgver=2023.11.1
pkgrel=1
arch=(any)
pkgdesc='Snapshot of the complete content of the KnotInfo and LinkInfo databases for Python'
url='https://github.com/soehms/database_knotinfo'
depends=(python)
makedepends=(python-build
             python-installer
             python-setuptools-scm
             python-wheel)
license=(GPL)
source=(https://pypi.io/packages/source/d/$_pyname/$_pyname-$pkgver.tar.gz)
sha256sums=('c58e6313905be537947308935056eea9e8e1f757edc01e89ad1ae0da3d91bd20')

build() {
  cd $_pyname-$pkgver
  python -m build --wheel --no-isolation
}

package() {
  cd $_pyname-$pkgver
  python -m installer --destdir="$pkgdir" dist/*.whl
}
